<commit_before>{ stdenv, fetchurl, cyrus_sasl, libevent }:

stdenv.mkDerivation rec {
  name = "libmemcached-1.0.18";
  
  src = fetchurl {
    url = https://launchpad.net/libmemcached/1.0/1.0.18/+download/libmemcached-1.0.18.tar.gz;
    sha256 = "10jzi14j32lpq0if0p9vygcl2c1352hwbywzvr9qzq7x6aq0nb72";
  };
  
  # Fix linking against libpthread (patch from Fedora)
  # https://bugzilla.redhat.com/show_bug.cgi?id=1037707
  # https://bugs.launchpad.net/libmemcached/+bug/1281907
  patches = [ ./libmemcached-fix-linking-with-libpthread.patch ];

  buildInputs = [ cyrus_sasl libevent ];

  meta = with stdenv.lib; {
    homepage = http://libmemcached.org;
    description = "Open source C/C++ client library and tools for the memcached server";
    license = licenses.bsd3;
    platforms = platforms.linux;
  };
}<commit_msg>libmemcached: build on darwin (patch from homebrew)<commit_after>{ stdenv, fetchurl, fetchpatch, cyrus_sasl, libevent }:

stdenv.mkDerivation rec {
  name = "libmemcached-1.0.18";

  src = fetchurl {
    url = https://launchpad.net/libmemcached/1.0/1.0.18/+download/libmemcached-1.0.18.tar.gz;
    sha256 = "10jzi14j32lpq0if0p9vygcl2c1352hwbywzvr9qzq7x6aq0nb72";
  };

  # Fix linking against libpthread (patch from Fedora)
  # https://bugzilla.redhat.com/show_bug.cgi?id=1037707
  # https://bugs.launchpad.net/libmemcached/+bug/1281907
  # Fix building on OS X (patch from Homebrew)
  # https://bugs.launchpad.net/libmemcached/+bug/1245562
  patches = stdenv.lib.optional stdenv.isLinux ./libmemcached-fix-linking-with-libpthread.patch
    ++ stdenv.lib.optional stdenv.isDarwin (fetchpatch {
      url = "https://raw.githubusercontent.com/Homebrew/homebrew/bfd4a0a4626b61c2511fdf573bcbbc6bbe86340e/Library/Formula/libmemcached.rb";
      sha256 = "1nvxwdkxj2a2g39z0g8byxjwnw4pa5xlvsmdk081q63vmfywh7zb";
    });

  buildInputs = [ cyrus_sasl libevent ];

  meta = with stdenv.lib; {
    homepage = http://libmemcached.org;
    description = "Open source C/C++ client library and tools for the memcached server";
    license = licenses.bsd3;
    platforms = platforms.linux ++ platforms.darwin;
  };
}
{'fraction_non_alphanumeric': 0.12699905926622765, 'fraction_numerical': 0.07055503292568203, 'mean_word_length': 3.992957746478873, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 8, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4224314', 'n_tokens_mistral': 916, 'n_tokens_neox': 816, 'n_words': 126}